

Índice

INTRODUCCION.....	5
1. CENTRADO DE TEXTOS.....	6
2. VISUALIZAR ESPACIOS EN BLANCO.....	7
3. EFECTOS MULTICOLORES EN LA PANTALLA	8
4. SUBRAYADO DE TEXTOS	9
5. COPY DE PANTALLA EN BAJA RESOLUCION.....	10
6. IMPRIMIR EN VIDEO INVERSO	11
7. ESCRITURA LETRA A LETRA CON SONIDO, EN VENTANA (MODE 2)	12
8. "PRESIONE UNA TECLA PARA CONTINUAR"	13
9 ENTRADA DE DATOS EN FORMA DE FICHAS	14
10. COPYCHR\$ PARA 464.....	15
11. CAT POR IMPRESORA (AMSDOS).....	16
12. CONVERSION DE UN 664 0 6128 EN UN 464.....	17
13. TABLA DE NÚMEROS DECIMALES Y HEXADECIMALES	18
14. COLORES PARPADEANTES.....	19
15. TECLAS DEFINIDAS PARA TRABAJAR EN BASIC	20
16. DIVISIBILIDAD.....	22
17. DEFINICION DE LA LETRA Ñ Y ñ	23
18. CONTROL DE SCROLL AUTOMÁTICO DE LA PANTALLA	24
19. DEFINICION DE ¿ y ¡.....	25
20. DEFINICIÓN DE NUEVOS CARACTERES Y SÍMBOLOS	26
21. PAUSE 0.....	28

35. BÚSQUEDA DE LETRAS EN UN TEXTO.....	46
36. BÚSQUEDA DE PALABRAS	47
37. DIAGRAMAS DE QUESO.....	49
38. DIAGRAMA DE BARRAS TRIDIMENSIONAL.....	51
39. IMPRESION TRANSPARENTE.....	52
40. CONTROL DE ERRORES.....	53
41. DESLIZAMIENTOS DE LOS ELEMENTOS DE UNA MATRIZ	54
42. ROTACIONES DE LOS ELEMENTOS DE UNA MATRIZ	56
43. ROTAR Y ENCAJAR.....	57
44. CLASIFICACIÓN ALFABÉTICA (Método REBOND).....	58
45. CLASIFICACION ALFABETICA (Método de la burbuja)	59
46. CLASIFICACION ALFABETICA (Método de SHELL-METZNER).....	61
APÉNDICE 1. EQUIVALENCIA ENTRE NÚMEROS DECIMALES Y HEXADECIMALES..	67
MENSAJES DE ERROR.....	68
OTROS TITULOS:.....	70
Revisión 2011.....	71

LOS TRUCOS DEL AMSTRAD

Autores:

A. BELLIDO

V. J. CAMPO

INTRODUCCION

En las páginas que siguen hemos reunido aquellos TRUCOS, rutinas e ideas de programación, de aplicación específica en el ordenador AMSTRAD, versiones CPC 464, 664 y 6128 y que nos han resultado útiles en nuestro trabajo en torno a este ordenador. Esperamos que su lectura y comprensión contribuya a hacer las cosas más fáciles y divertidas al programador de AMSTRAD.

Algunos TRUCOS como verá son específicos del AMSTRAD, e incluso en algún caso, los menos, propios de alguno de los modelos, otros sin embargo son de índole más general y fácilmente implantables en otros ordenadores. La compatibilidad entre los tres modelos es casi total, excepto obviamente en lo que se refiere a discos y se ha tratado de mejorar esta compatibilidad con alguno de los TRUCOS expuestos.

El libro está concebido como un manual de consulta, al cual podamos remitirnos, buscando algún tipo de aplicación difícil de encontrar en un manual de uso o en libros específicos, que si bien pueden contener información, ésta no está catalogada y por tanto difícil de encontrar en un momento dado.

Cada TRUCO va incluido en una ficha y ésta se divide en cuatro partes. La primera, "EFECTO", describe los efectos conseguidos con la aplicación del mismo.

La segunda, "CÓMO CONSEGUIRLO", incluye la rutina en BASIC o código máquina que permite conseguir el efecto descrito.

La tercera, "EJEMPLO", incluye un ejemplo sencillo que ilustra como se usa la rutina descrita anteriormente. Generalmente, para utilizar el ejemplo es necesario haber cargado en memoria previamente la rutina.

La cuarta parte y última, "EXPLICACION", trata de describir "cómo y porqué" funciona y se consigue el efecto. De esta explicación podrán conseguirse en muchos casos ideas para nuevas aplicaciones, siendo este uno de los objetivos, no tratamos sólo exponer una colección de TRUCOS sino facilitar nuevos caminos al programador e indicarle mayores posibilidades en la resolución de problemas específicos.

La numeración de las rutinas comienza en la línea 9000 a fin de que no interfieran en sus programas BASIC, no obstante puede reenumerarlas a su gusto o necesidades e incluso utilizarlas directamente y no como subrutinas.

Si lo desea puede grabarlas por separado y cargarlas desde la cinta o disco en que las almacene, en sus programas mediante la instrucción MERGE, reenumerando posteriormente el conjunto.

El asterisco que figura bajo el número de ficha, indica que el TRUCO es válido para cualquiera de los tres modelos de AMSTRAD, en caso contrario se indica en qué modelos puede utilizarse.

Por último, indicarle que para la explicación de estos TRUCOS no necesita tener unos profundos conocimientos del BASIC del AMSTRAD, aunque evidentemente su total comprensión se verá facilitada por su experiencia.

1. CENTRADO DE TEXTOS


EFECTO

Centra cualquier texto en la pantalla, en cualquier MODE, de forma que los mismos aparecen simétricamente respecto al eje central.

CÓMO CONSEGUIRLO

Utilice la rutina que indicamos a continuación, cambiando el número 20 por los siguientes valores, dependiendo del modo de pantalla.

MODE 0	10
MODE 1	20
MODE 2	40



```
9000 REM CENTRADO EN 40 COLUMNAS
9010 DEF FN T(T$)=20-(LEN (T$)/2)
9020 PRINT TAB(FN T(A$))A$
9030 RETURN
```

EJEMPLO

Pruebe el efecto que se consigue con el ejemplo que indicamos para MODE 1.

```
10 CLS
20 PRINT"1234567890123455789012345675901234567890"
30 A$="CENTRADO DE TEXTOS"
40 GOSUB 9000
50 A$="40 columnas"
60 GOSUB 9000
70 END
```

EXPLICACION

En la línea 9010, definimos una función que determina la posición del tabulador donde debe comenzar la impresión y que depende de la longitud del texto y del modo de pantalla en que estamos trabajando.

NOTA de la revisión 2011

Los colores en los fuentes no existían en el libro original, y no se apreciarán si se teclea dichos fuentes en un CPC real o en un emulador. Estos colores se incluyen como ayuda para la detección de errores.

De igual modo, las capturas de pantalla no aparecían en el libro original.

2. VISUALIZAR ESPACIOS EN BLANCO

EFECTO

Muestra los espacios en blanco contenidos en un listado en vídeo inverso.

CÓMO CONSEGUIRLO

Introducir la rutina en forma de programa o bien como comandos directos (sin números de línea).

SYMBOL AFTER 1

KEY 0,"symbol 32,0"+CHR\$(13)


KEY 1,"symbol 32,255,255,255,255,255,255,255"+CHR\$(13)

EJEMPLO

Utilice cualquier programa, lístelo y observe el efecto obtenido.

EXPLICACIÓN

Pulsando la tecla [f1] (teclado numérico auxiliar), los espacios se hacen visibles en video inverso, esto le permitirá suprimir todos los espacios innecesarios que se introdujeron al teclear el programa, sobre todo al utilizar la tecla COPY. Para volver a la normalidad, pulsar [f0].



3. EFECTOS MULTICOLORES EN LA PANTALLA

EFECTO

Las tres rutinas que presentamos producen un efecto multicolor sobre la pantalla, tipo explosión. La primera afecta sólo al borde, la segunda al cuadro de la pantalla y la tercera a toda la pantalla.



CÓMO CONSEGUIRLO

```
9000 REM Efecto de borde
9010 duracion=200
9020 FOR k=1 TO duracion
9030 var=RND*50+50
9040 OUT &7F00,50
9050 OUT &7F00,var
9060 NEXT k
9070 BORDER 1
9080 RETURN
```

```
9000 REM Efecto en pantalla
9010 duracion=200
9020 FOR k=1 TO duracion
9030 var=RND*50+50
9040 OUT &7F00,0
9050 OUT &7F00,var
9060 NEXT k
9070 BORDER 1
9080 RETURN
```

```
9000 REM Efecto en toda la pantalla
9010 duracion=200
9020 FOR k=1 TO duracion
9030 var=RND*50+50
9040 OUT &7F00,0:OUT &7F00,var
9050 OUT &7F00,50:OUT &7F00,var
9060 NEXT k
9070 BORDER 1
9080 RETURN
```

EJEMPLO

Utilice cualquiera de sus programas en el que desee alguno de estos efectos y llámela mediante la instrucción GOSUB 9000.

EXPLICACIÓN

La duración del efecto puede modificarse, cambiando el valor en la línea 9010. Pruebe para valores distintos del RND en la 9030.

4. SUBRAYADO DE TEXTOS

EFECTO

Esta rutina produce el subrayado de cualquier texto, sobreimprimiendo el símbolo "_" en la misma línea que el texto.

CÓMO CONSEGUIRLO

Utilice la rutina.

```
9000 REM SUBRAYADO
9010 PRINT TAB(N)A$;CHR$(22);CHR$(1)
9020 PRINT CHR$(11);
9030 FOR L=1 TO LEN (A$):PRINT "_";:NEXT
9040 RETURN
```

EJEMPLO

Pruebe este sencillo programa junto con la rutina anterior.

```
10 a$="SUBRAYADO DE TEXTOS"
20 GOSUB 9000
30 END
```

CÓMO CONSEGUIRLO

CHR\$(22) es una opción de transparencia, lleva un parámetro que puede valer 0 (desactivado) ó 1 (activado) Así CHR\$ (22); CHR\$ (1) produce una sobreimpresión, respetando lo anterior (ver ficha 39).

CHR\$ (11) hace subir el cursor una línea.

El bucle de la 9030 imprime el símbolo de subrayar tantas veces como caracteres tenga el texto A\$.

5. COPY DE PANTALLA EN BAJA RESOLUCION

* (para 464 ver ficha 10) 664.6128

EFECTO

Realiza un COPY de pantalla en baja resolución, es decir sólo para textos, en cualquier modo (0, 1 o 2).

Esta rutina le será más útil, que tener que duplicar los PRINT en el caso de querer obtener los resultados por pantalla e impresora. Prepare únicamente la salida hacia pantalla como le guste y obtendrá en papel una copia exacta.

CÓMO CONSEGUIRLO

```
9000 REM COPY DE PANTALLA PARA 40 COLUMNAS
9010 WIDTH 40
9020 FOR F=1 TO 25
9030 FOR C=1 TO 40
9040 LOCATE C,F
9050 A$=COPYCHR$(#0)
9060 PRINT #8,A$
9070 NEXT C,F
```

EJEMPLO

```
10 CLS:FOR K=1 TO 50
20 F=INT(RND*25)+1
30 C=INT(RND*40)+1
40 LOCATE C,F:PRINT"A"
50 NEXT K
60 GOSUB 9000
70 END
```

EXPLICACIÓN

La rutina utiliza la instrucción COPYCHR\$ (disponible para los modelos 664 y 6128), lee el carácter que hay en la posición actual del cursor, lleva forzosamente un parámetro que indica la ventana sobre la que ha de actuar, si no se han indicado ventanas debe llevar 0 que es la normal.

La línea 9010 fija el número de caracteres por línea para la impresora, si se desea trabajar en MODE 0 ó 2, cambiar el valor 40 por 20 y 80 respectivamente.

La línea 9020 establece un bucle entre la línea 1 y la 25 y por tanto no depende del modo de pantalla.

La 9030 establece un bucle entre la columna 1 y la 40, para MODE 0 el limite superior será 20 y para MODE 2,80.

La 9040 sitúa el cursor y la 9050 lee el contenido y lo almacena en la variable A\$.

La línea 9060 imprime el contenido de A\$ por la impresora (8). El punto y coma hace que no se produzca el salto de línea (retorno de carro), dicho salto sólo se realiza cuando el número de columnas impresas es igual a las indicadas por la orden WIDTH de la línea 9010.

6. IMPRIMIR EN VIDEO INVERSO

EFECTO

Resalta los datos en pantalla apareciendo en vídeo inverso, es decir intercambia los colores de tinta y papel.

CÓMO CONSEGUIRLO

Utilice el carácter de control 24 en la forma CHR\$ (24) en sus PRINT, en la forma que se indica en el ejemplo.

Si desea cambiar toda la pantalla introduzca.

```
PRINT CHR$ (24):CLS + ENTER
```

Para restituir las condiciones iniciales, repita la orden.

EJEMPLO

```
10 PRINT "VIDEO NORMAL"  
20 PRINT CHR$ (24); "VIDEO INVERSO"; CHR$ (24)
```

El primer CHR\$ (24) invierte los colores y el segundo restaura las condiciones iniciales.
NOTA: Los puntos y coma de la línea 20 pueden omitirse.

EXPLICACIÓN

El carácter 24, cuyo nombre es CAN, produce el intercambio entre los colores de la tinta y el papel, no lleva ningún parámetro por tanto se utiliza como CHR\$ (24), sólo o actuando sobre PRINTs concretos.

7. ESCRITURA LETRA A LETRA CON SONIDO, EN VENTANA MODE 2!

EFECTO

Dado un texto de cualquier longitud, lo imprime letra a letra de forma más o menos rápida, añadiéndole un efecto sonoro. Permite enviar mensajes móviles en ventana en el MODE 2 (80 columnas).

CÓMO CONSEGUIRLO

```
9000 REM RUTINA DE ESCRITURA EN VENTANA CON SONIDO (80 COLUMNAS)
9500 MODE 2
9505 PRINT CHR$(24)
9525 LOCATE 5,22: PRINT
"-----"
----"
9540 FOR K=1 TO LEN(T$)
9550 LOCATE 1,24: PRINT MID$(T$,K,79)
9560 SOUND 1,3,10 :      FOR Y=1 TO 60: NEXT Y
9570 NEXT K
9575 PRINT CHR$(24)
9580 RETURN
```

EJEMPLO

Compruebe con este programa y la rutina anterior, el efecto conseguido

```
10 T$="          ESTE TEXTO CORRESPONDE A UN EJEMPLO DE ESCRITURA EN VENTANA
CON MOVIMIENTO CONTINUO, LA FRAGMENTACION DE LA CADENA SE REALIZA CON MID$ Y
LA ESCRITURA CON UN LOCATE FIJO, DONDE SE UBICA          "
20 GOSUB 9000
30 CLS
40 END
```

EXPLICACIÓN

La impresión se realiza mostrando 79 caracteres del total del texto, mediante la instrucción de fragmentación MID\$ (línea 9550) donde T\$ es la variable que contiene el texto, K la variable del bucle que varía entre 1 y la longitud de T\$ y 79 es el número de caracteres de la subcadena que se muestra.

La línea 9560 produce un sonido en cada PRINT y una pequeña pausa.

Si no se desea borrar la pantalla suprima la línea 9500.

Para vídeo normal, suprimir las líneas 9505 y 9575.

Si desea este efecto en MODE 1, sustituya el valor 79 de la línea 9550 por 39 y modifique la 9500. Recuerde que T\$ no puede contener más de 255 caracteres.

". #PRESIONE UNA TECLA PARA CONTINUAR#

EFEECTO

En muchos programas, antes de continuar es necesario un control o parada del mismo. El mensaje clásico es "Presione una tecla para continuar".

En algunos ordenadores la sentencia PAUSE 0 permite detener indefinidamente un programa. La rutina que presentamos a continuación produce este efecto y el de presentar la frase citada en forma de texto móvil.

CÓMO CONSEGUIRLO

Introduzca esta rutina en su ordenador.

```
9000 REM TEXTO MOVIL + PAUSA
9010 REM PRESIONE UNA TECLA PARA CONTINUAR
9020 TEXTO$="    PULSE UNA TECLA PARA CONTINUAR    "
9030 S$=TEXTO$
9040 PRINT CHR$(24);
9050 FOR CON=1 TO 40
9060 LOCATE 1,25
9070 PRINT TEXTO$;
9080 TEXTO$=MID$(S$,CON)+LEFT$(S$,CON)
9090 CON$=INKEY$:IF CON$<>" " THEN 9110
9100 NEXT CON: PRINT CHR$(7): GOTO 9050
9110 PRINT CHR$(24):CLS: RETURN
```

EJEMPLO


Utilice uno de sus programas o utilice estas líneas para ver el efecto

```
10 CLS
20 GOSUB 9000
30 END
```

EXPLICACIÓN

La línea 9090 produce el efecto de pausa citado, si se pulsa una tecla borra la pantalla, restaura los colores y vuelve al programa principal (línea 9110), en caso contrario continúa visualizándose el texto.

Cada vez que termina de imprimirse un texto se produce un BEEP (pitido) por la acción del carácter CHR\$(7).



\$ ENTRADA DE DATOS EN FORMA DE FICHAS

EFECTO

En ocasiones y con objeto de facilitar la entrada de datos conviene presentar estos en pantalla en forma de ficha, de manera que los INPUTS aparezcan en las casillas convenientes de la ficha.

CÓMO CONSEGUIRLO

Sobre una plantilla cuadriculada que simule la pantalla, en el modo deseado (ver plantillas en el apartado "Para su referencia..." de su manual). Sitúe los textos de los INPUTS que formarán la ficha, en la forma que desee, saque los valores de columna y fila de comienzo de cada texto e introduzca en líneas DATA las triadas de valores.

DATA columna, fila, "Texto del INPUT".

Calcule además los valores de columna y fila donde deben aparecer el cursor en los INPUTs, e introdúzcalos también en líneas DATA.

Siga luego el ejemplo que proponemos y le será fácil adaptarlo a sus necesidades.

EJEMPLO

```
10 REM INPUTS EN FICHAS
20 DATA 1,3,NOMBRE,1,5,CALLE
30 DATA 25,5,No.,1,7,POBLACION,25,7,TEL
40 DATA 8,3,8,5
50 DATA 28,5,11,7,29,7
60 CLS
70 A$=STRING$(40,"#")
80 PRINT A$
90 LOCATE 1,9: PRINT A$
100 PRINT CHR$(24)
110 FOR K=1 TO 5
120 READ C,F: LOCATE C,F
130 READ A$: PRINT A$
140 NEXT K
150 PRINT CHR$(24)
160 FOR K=1 TO 5
170 READ C,F: LOCATE C,F
180 INPUT "",A$(K)
190 NEXT K
```



EXPLICACIÓN

Las líneas DATA se obtienen en la forma indicada anteriormente.

La línea 70 produce una cadena de 40 caracteres iguales a #

Las líneas 80 y 90 sólo son de adorno para formar la ficha.

La línea 100 produce un vídeo inverso (intercambia los colores de la tinta y el papel). El primer bucle (110 a 140) imprime los textos de los INPUTS.

El segundo sitúa el cursor en los sitios prefijados, de forma que los INPUTS se producen en dichos sitios.

En la línea 180, si se suprime "", aparece el símbolo ? dentro del cursor.

1&. COPYC%R ' PARA 464

EFECTO

Permite obtener en el 464 un comando análogo al COPYCHR\$ de los modelos de disco.

CÓMO CONSEGUIRLO

Cargue la rutina en código máquina con el cargador BASIC que se acompaña (compruebe bien los valores de los DATA antes de hacer RUN).

Una vez instalado tiene disponible su función. Inicialice una variable alfanumérica con un espacio (por ejemplo X\$ = " ") y llame a la rutina mediante CALL HIMEM + 1, X\$

```
10 MEMORY (HIMEM-50)
20 s=1
30 READ a$
40 WHILE a$<>"*"
50 POKE HIMEM+s,VAL("&" +a$)
60 s=s+1:READ a$
70 WEND
80 DATA CD,78,BB,E5,DD,7E,08,CD : REM dos ultimos dudosos
90 DATA B4,BB,F5,CD,78,BB,CD,87
100 DATA BB,CD,75,BB,CD,60,BB,FE
110 DATA C0,28,0B,DD,68,03,DD,66 : REM tres ultimos dudosos
120 DATA 03,23,5E,23,56,12,F1,CD : REM dos ultimos dudosos
130 DATA B4,BB,E1,CD,75,BB,C9,*
140 END
```

EJEMPLO

Equivalencias entre COPYCHR\$ y la llamada a la rutina

664 ó 6128	464
LOCATE 10,5: X\$ = COPYCHR\$	CALL HIMEM + 1, X\$
LOCATE #1, 15, 15 : X\$ = COPYCHR\$ (#1)	CALL HIMEM + 1, X\$, 1

De esta forma podrá traducir listados programa del 664 ó 6128 a su 464, que utilicen esta instrucción.

NOTA de la revisión 2011

Algunos de los DATA del libro original no se leían correctamente, y puede existir algún error en las cifras finales de algunas líneas.

11. CAT POR IMPRESORA AMSDOS!

664-6128

EFECTO

Produce sobre la impresora el mismo efecto que CAT sobre la pantalla. Permite por tanto listar por impresora el contenido de los archivos en disco y simultáneamente por pantalla.

Aunque el método es válido para archivos en cinta, no recomendamos su utilización habida cuenta de la diferencia de la sentencia CAT sobre disco y sobre cinta y por no disponer el 464 de la función COPYCHR\$ utilizada en esta rutina, (ver ficha 10).

CÓMO CONSEGUIRLO

```
9000 REM CAT PARA IMPRESORA
9010 MODE 1:WIDTH 40
9020 CAT
9030 PRINT "#"
9040 FOR F= 1 TO 25
9060 FOR C=1 TO 40
9060 LOCATE C,F:A$=COPYCHR$(#0)
9070 IF A$="#" THEN 9090 ELSE PRINT #8,A$;
9080 NEXT C,F
9090 END
```

EJEMPLO

Introduzca uno de sus discos, encienda la impresora y teniendo el programa anterior en memoria (o salvado en disco) utilice RUN y obtendrá un catálogo por impresora.

EXPLICACIÓN

La línea 9010 prepara la pantalla y la impresora para trabajar en 40 columnas.

9020 produce un catálogo del contenido del disco sobre la pantalla.

9030 escribe en pantalla un símbolo de control para que la copia se reduzca al número de líneas estricto que ocupa el catálogo.

9050 a 9080 bucle idéntico a la rutina de la ficha 5, añadiendo el control del carácter4 producido por la línea 9030 como fin de la impresión.

Esta rutina es válida bajo el sistema operativo AMSDOS, que es el inicial en su AMSTRAD. Si se encuentra bajo CPM resulta aún más fácil (ver ficha 23).

12. CONVERSION DE UN 664 & 612" EN UN 464

664-6128

EFECTO

No se trata de un paso atrás, sino al contrario permitir a los modelos 664 y 6128 el uso de programas concebidos exclusivamente para el 464.

CÓMO CONSEGUIRLO

Introduzca esta rutina y antes de hacer RUN, sálvela en disco, compruebe detalladamente los valores de las líneas DATA y sólo cuando esté seguro salve la versión definitiva concluyendo la línea 9095 NEW que borrará el programa.

```
9000 REM CONVERTIR EN UN CPC 464
9010 MEMORY &7FF
9020 ORI=&8000
9030 FOR K=0 TO 62
9040 READ CON
9050 POKE ORI+K, CON
9060 S=S+CON
9070 NEXT K
9080 IF S<>6058 THEN PRINT "ERROR EN LOS DATOS": END
9090 CALL &8000
9095 NEW
9100 DATA 205,200,188,122,179,32,5,6,0,17,6,192,237,83,60,128
9110 DATA 237,67,62,128,33,28,128,14,252,205,22,189,42,60,128
9120 DATA 237,75,62,128,72,62,201,50,203,188,34,57,128,121,50,59,128
9130 DATA 17,64,0,33,255,171,223,57,128,0,0,0,0,0,0,0
```

EXPLICACIÓN

Los modelos 664 y 6128 necesitan 1284 bytes para la gestión del BASIC de forma que al encender si usamos la orden PRINT HIMEM veremos que nos da el valor 42619, sin embargo en un 464 obtenemos 43903, esta diferencia es la ocupada por el sistema operativo del disco.

Con el fin de que el BASIC recupere toda la memoria guarda la rutina en un disco con SAVE "464" y antes de introducir los programas comerciales para 464, que le darían el error "MEMORY FULL", haga RUN "464". Cárguelos ahora. Verá como se trata de un paso adelante y no hacia atrás.

13. TABLA DE N(MEROS DECIMALES Y %EXADECIMALES

EFEECTO

Produce una tabla de equivalencias entre números decimales y hexadecimales lo que le facilitará la conversión entre ambos sistemas de numeración. Al final se expone la tabla obtenida con este programa. (Apéndice 1).

#####

CÓMO CONSEGUIRLO

```
10 PRINT #8,"      EQUIVALENCIA ENTRE NUMEROS DECIMALES Y HEXADECIMALES"
20 PRINT #8,"      ====="
30 PRINT #8
40 WIDTH 78:MODE 2
50 FOR K=1 TO 55
60 PRINT K;"=";HEX$(K,2),K+50;"=";HEX$(K+50,2),K+100;"=";HEX$
(K+100,2),K+150;"=";HEX$(K+150),K+200;"=";HEX$(K+200)
70 PRINT #8, K;"=";HEX$(K,2),K+50;"=";HEX$(K+50,2),K+100;"=";HEX$
(K+100,2),K+150;"=";HEX$(K+150),K+200;"=";HEX$(K+200)
80 NEXT
```

Si no dispone de impresora elimine las líneas 10, 20, 30 y 70, modifique la 40 de forma 40 MODE 2 y obtendrá la tabla sólo por pantalla. Si lo desea puede consultar la misma en el apéndice del libro.

EXPLICACIÓN

El programa aprovecha la facilidad dada por la sentencia HEX\$, esta lleva dos parámetros, el primero es el número decimal a convertir y el segundo el número de dígitos con que queremos expresar el número hexadecimal resultante, en nuestro caso 2 puesto que el bucle calcula números entre 1 y 255, que en hexadecimal equivale a 00 y FF respectivamente.

Recuerde que la numeración hexadecimal (base 16) se compone de 16 símbolos) 1, 2, 3, 4, 5, 6, 7, 8, 9, A, B, C, D, E, F.

cuyos valores decimales correspondientes son

1, 2, 3, 4, 5, 6, 7, 8, 9, 10, 11, 12, 13, 14, 15.

Así pues como los 9 primeros son idénticos, sólo tiene que recordar los valores de las letras.

Para pasar rápidamente a decimal, multiplique el primero (de la izquierda) por 16 y súmele el de la derecha por ejemplo.

FA = $15 \cdot 16 + 10 = 240 + 10 = 250$ (decimal)

3B = $3 \cdot 16 + 11 = 48 + 11 = 59$ (decimal)

Si el número hexadecimal tuviera más dígitos, la regla general es:

$4BF1 = 4 \cdot 16^3 + 11 \cdot 16^2 + 15 \cdot 16^1 + 1 \cdot 16^0 = 4 \cdot 4096 + 11 \cdot 256 + 15 \cdot 16 + 1$

14. COLORES PARPADEANTES

EFFECTO

Parpadeo de los colores de la pantalla y/o el borde.

CÓMO CONSEGUIRLO

Pantalla: Al asignar los colores mediante la sentencia INK, introduzca 2 colores en vez de 1, por ejemplo INK 1, 10, 16.

Border: Igualmente introduzca 2 colores en esta sentencia, por ejemplo pruebe BORDER 10, 20.

En ambos casos la velocidad del parpadeo puede modificarse mediante SPEED INK (n1, n2).

n1 periodo del primer color (1 ud = 0,02 segundos).

n2 período del segundo color (1 ud = 0,02 segundos).

EJEMPLO

Pruebe usted mismo con combinaciones y velocidades distintas y elija para sus programas aquellos más agradables.

Según indica el manual, existen combinaciones hipnotizantes.

15. TECLAS DEFINIDAS PARA TRABAJAR EN BASIC

EFECTO

Convierte algunas de las teclas del teclado alfabético en las sentencias de BASIC más comunes en programación. Para obtener cualquiera de las indicadas sólo tiene que apretar CONTROL y simultáneamente la tecla seleccionada.

CÓMO CONSEGUIRLO

Teclee y salve esta rutina y antes de programar llámela mediante RUN, lleva una instrucción de autodestrucción (línea 180), que no debe de ser incluida hasta que esté seguro de que funciona.

-- Revisar: alguna tecla cambia sin tener que hacerlo: T → D, etc --

```
5 REM TECLAS DEFINIDAS - TECLADO ALFABETICO
```

```
10 KEY DEF 36,0,108,76,141
15 KEY 141,"LIST "
20 KEY DEF 27,0,112,80,142
25 KEY 142," PRINT "
30 KEY DEF 60,0,115,83,143
35 KEY 143 , "SAVE "+CHR$(34)
40 KEY DEF 69,0,97,65,144
45 KEY 144 , "LOAD "+CHR$(34)
50 KEY DEF 52,0,103,71,145
55 KEY 145," GOTO "
60 KEY DEF 53,0,102,70,146
65 KEY 146,"FOR "
70 KEY DEF 51,0,116,84,147
75 KEY 147," TO "
80 KEY DEF 46,0,110,78,148
85 KEY 148,"NEXT "
90 KEY DEF 50,0,114,82,149
95 KEY 149, "RUN "+CHR$(13)
100 KEY DEF 35,0,105,73,150
105 KEY 150,"INPUT "
110 KEY DEF 50,0,101,69,151
115 KEY 151,"EDIT "
120 KEY DEF 62,0,99,67,152
125 KEY 152,"CHR$("
130 KEY DEF 34,0,111,79,153
135 KEY 153, "LOCATE "
140 KEY DEF 43,0,121,89,154
145 KEY 154,"IF "
150 KEY DEF 51,0,116,84,155
155 KEY 155," THEN "
160 KEY DEF 42,0,117,85,156
165 KEY 156," USING "
170 KEY DEF 51,0,100,68,157
175 KEY 157, "DATA "
180 CLS: NEW
```

Si lo desea puede cambiar las instrucciones asignadas a cada letra.

LETRA y EQUIVALENCIA

L - LIST

P - PRINT

S - SAVE"
L - LOAD"
G- GOTO
F - FOR
N - NEXT
R - RUN + ENTER
I - INPUT
E - EDIT
C - CHR\$(
O - LOCATE
Y - IF
T - THEN
U - USING
D - DATA

16. DIVISIBILIDAD

EFECTO

Calcula si un número es divisible entre otro, es decir si el resto de su cociente es cero.

CÓMO CONSEGUIRLO

Basta con verificar si su cociente es igual a la parte entera de dicho cociente.

IF (n1 / n2) = INT (n1 / n2) SÍ son divisibles

si no, no lo son

siendo n1 y n2 los números

EJEMPLO

```
10 INPUT "Divisibles por"; n2
20 FOR n1 = 1 TO 100
30 IF (n1 / n2) = INT (n1 / n2) THEN PRINT n1
40 NEXT n1
```

Este programa obtiene todos los números comprendidos entre 1 y 100 que son divisibles por n2.


EXPLICACIÓN

Este TRUCO es una aplicación simple de la instrucción INT que devuelve la parte entera de un número (sin los decimales), de manera que si dos números son divisibles el cociente será entero (resto cero) y por tanto igual a la parte entera de dicho cociente.

NOTA de la revisión 2011

Se puede conseguir el mismo efecto usando la instrucción MOD, que devuelve el resto de una división. Si dicho resto es 0, el primer números es divisible entre el segundo:

IF n1 MOD n2 = 0



17. DEFINICION DE LA LETRA) Y *

EFECTO

Si su modelo de AMSTRAD no posee la letra, no se preocupe: podemos "fabricarla".

CÓMO CONSEGUIRLO

Con esta pequeña rutina será suficiente para cambiar letras

```
9000 SYMBOL 240,60,0,216,102,102,102,102,0
9010 SYMBOL 241,124,0,230,246,222,198,198,0
9020 RETURN
```

EJEMPLO


Pruebe este ejemplo con la rutina anterior y verá el efecto de las letras. Si no le gusta su forma no le será muy difícil cambiarlas. Dibújelas en papel cuadriculado y saque los valores correspondientes.

```
10 REM NUEVAS LETRAS
20 GOSUB 9000
30 PRINT"FELIZ A";CHR$(241);"O NUEVO"
40 PRINT"Buen vino a";CHR$(240);"ejo"
50 END
```

EXPLICACIÓN

La instrucción SYMBOL nos permite definir nuevos caracteres o símbolos. Disponemos en principio de los caracteres 240 a 255 y de ellos hemos elegido los dos primeros, si necesitásemos más de los 16 citados debemos usar la instrucción SYMBOL AFTER previamente a fin de hacer la reserva necesaria.

Detrás del código, sigue una lista de 8 números, que son los valores decimales del dibujo de las letras (ver TRUCO 20 definición de nuevos caracteres y símbolos).



1". CONTROL DE SCROLL AUTOM+TICO DE LA PANTALLA

EFECTO

Controla el movimiento de scrolling de la pantalla, de forma que los PRINT aparecen en bloques de líneas controladas por la presión de una tecla.

CÓMO CONSEGUIRLO

Basándonos en el TRUCO 16 podemos controlar el número de líneas en pantalla y su parada mediante

IF contador/nº líneas = INT (contador / nº líneas) parar
si no, continua siguiente PRINT


EJEMPLO

Con este ejemplo podrá ver el efecto claramente, el contador es la variable K del bucle y el número de líneas establecido es 20.

```
10 REM PARADA DEL SCROLL
20 CLS
30 PRINT"NUMERO      CUADRADO      CUBO"
40 PRINT"===== "
50 FOR K=1 TO 100
60 PRINT K,K^2,K^3
70 IF (K/20)=INT(K/20) THEN GOSUB 90
80 NEXT
85 END
90 Z$=INKEY$:IF Z$="" THEN 90
100 CLS
110 PRINT"NUMERO      CUADRADO      CUBO"
120 PRINT"===== "
130 RETURN
```

EXPLICACIÓN

El bucle (50-80) imprime los 100 primeros números con sus cuadrados y cubos. Al llegar a 20 se verificará el IF de la línea 70 y por tanto salta a la subrutina 90, en la que INKEY\$ produce una espera en tanto no se apriete una tecla; en este caso se borra la tabla, se vuelve a imprimir la cabecera y se retorna al bucle continuando por el valor de K = 21 y así sucesivamente con paradas cada 20 números.



NUMERO	CUADRADO	CUBO
1	1	1
2	4	8
3	9	27
4	16	64
5	25	125
6	36	216
7	49	343
8	64	512
9	81	729
10	100	1000
11	121	1331
12	144	1728
13	169	2197
14	196	2744
15	225	3375
16	256	4096
17	289	4913
18	324	5832
19	361	6859
20	400	8000

Pruebe con otros valores en lugar de 20 y vea el efecto.

1\$. DEFINICION DE , - .

EFECTO

En el TRUCO 17 definimos Ñ y ñ. Completando los caracteres propios del castellano definiremos ahora ¿ y ¡.

CÓMO CONSEGUIRLO

Esta rutina puede añadirse a la expuesta en el TRUCO 17, si la tiene archivada, teclee esta, haga un MERGE y tendrá ambas juntas y disponibles (la numeración de líneas lo permite).

```
9020 SYMBOL 242,0,24,0,24,48,46,102,60: ' ABRIR INTERROGACION
9030 SYMBOL 243,24,0,24,24,24,24,24,0: ' ABRIR ADMIRACION
9040 RETURN
```

EJEMPLO

Imprima el texto que desee añadiendo los nuevos caracteres mediante CHR\$ (242) y CHR\$(243), en la misma forma que lo hacíamos en el TRUCO 17:

```
10 GOSUB 9020:PRINT CHR$(242); CHR$(243):END
```

EXPLICACIÓN

Ver TRUCOS 17 y 20.



2&. DEFINICI / N DE NUEVOS CARACTERES Y SÍMBOLOS

EFFECTO

Permite obtener rápidamente los valores decimales de un gráfico definido por el usuario, para su utilización en la sentencia SYMBOL.

CÓMO CONSEGUIRLO

1º) Dibuje una cuadrícula de 8 por 8 cuadros y comience a numerar de derecha a izquierda las columnas con la siguiente secuencia de números 1, 2, 4, 8, 16, 32, 64, 128.

2º) Rellene con negro los cuadros que desee, hasta formar su dibujo.

3º) Calcule el valor de cada fila (comenzando por la superior) teniendo en cuenta que a un cuadro negro le corresponde el valor de la columna en que se encuentra y a uno blanco nada, si hay más de un cuadro negro en una fila sume los valores de cada uno.

4º) Utilice tras la sentencia SYMBOL n.º carácter al que se asigna el dibujo, valor de la primera fila, valor de la 2.ª..., valor de la 8.ª.

Así de fácil.

EJEMPLO

128	64	32	16	8	4	2	1	
								0
			■	■	■			$4 + 8 + 16 = 28$
		■	■	■	■	■		$2 + 4 + 8 + 16 + 32 = 62$
	■	■		■		■	■	$1 + 2 + 8 + 32 + 64 = 107$
	■	■		■		■	■	$= 107$
	■	■	■	■	■	■	■	$1 + 2 + 4 + 8 + 16 + 32 + 64 = 127$
	■			■			■	$1 + 8 + 64 = 73$
	■						■	$1 + 64 = 65$

Teclee

SYMBOL 240, 0, 28, 62, 107, 107, 127, 73, 65

Pruebe ahora

PRINT CHR\$(240)

y obtendrá un "COCO" en pantalla.

EXPLICACIÓN

Si sustituimos en la cuadrícula anterior los cuadros negros por unos y los blancos por otros, obtendremos una representación binaria del gráfico, de forma que lo que hemos hecho anteriormente es pasar de binario a decimal, de forma rápida y sin recurrir a fórmulas complicadas.



0	0	0	0	0	0	0	0
0	0	0	1	1	1	0	0
0	0	1	1	1	1	1	0
0	1	1	0	1	0	1	1
0	1	1	0	1	0	1	1
0	1	1	1	1	1	1	1
0	1	0	0	1	0	0	1
0	1	0	0	0	0	0	1

Tenga en cuenta que por ejemplo la segunda fila que en binario es: 000 111 00 para pasarlo a decimal, hay que coger comenzando por la derecha cada uno de los números y multiplicarlo por la base (2) elevada al lugar que ocupa y por último sumar todo, es decir

Binario 0 0 0 1 1 1 0 0

$$0 \times 2^0 = 0$$

$$0 \times 2^1 = 0$$

$$1 \times 2^2 = 4$$

$$1 \times 2^3 = 8$$

$$1 \times 2^4 = 16$$

$$0 \times 2^5 = 0$$

$$0 \times 2^6 = 0$$

$$0 \times 2^7 = 0$$

$$= 28 \text{ decimal}$$


Seguro que de la otra forma le resulta menos complicado y rápido.

21. PAUSE &

EFECTO

Si ha utilizado otros ordenadores, como por ejemplo el Spectrum, habrá observado que algunos disponen de una sentencia PAUSE cuyo parámetro si es 0 produce una pausa por tiempo indefinido, hasta que una tecla es pulsada.

Su mayor aplicación está en la combinación con el clásico mensaje "PULSE UNA TECLA PARA CONTINUAR", que permite visualizar una pantalla durante el tiempo que deseemos. El AMSTRAD no dispone de esta sentencia, pero no se preocupe es muy fácil de emular.



CÓMO CONSEGUIRLO

En la ROM de su ordenador tiene la solución, basta con llamar a la dirección &BB18 hexadecimal, mediante la orden CALL (llamada a rutinas en código máquina). Introduzca pues en sus listados en vez de PAUSE 0, CALL &BB18 y obtendrá el mismo efecto.

También puede obtenerse dicho efecto con CALL & BB06.

EJEMPLO

```
9000 REM RUTINA DE MENSAJE DE CONTINUAR.  
9010 WINDOW #1, 1, 40, 25, 25  
9020 PRINT #1, "PRESIONE UNA TECLA PARA CONTINUAR"  
9030 CALL &BB18  
9040 RETURN
```

Esta rutina está pensada para trabajar en MODE 1 (normal), si la desea en MODE 2 modifique la línea 9010 WINDOW#1, 1, 80, 25, 25 ó en MODE 0 por 9010 WINDOW#1, 1, 20, 25, 25. El texto de la línea 9020 deberá también modificarlo consecuentemente, alargándolo o acortándolo.

22. NUEVO JUEGO DE CARACTERES

EFEECTO


Produce un nuevo juego de caracteres completo, sin tener que recurrir a su definición uno a uno mediante la orden SYMBOL.

CÓMO CONSEGUIRLO

Aprovechando una rutina de la propia ROM, utilice dentro de su programa la orden CALL &BA0A en el momento en que lo desee.

EJEMPLO

```
10 MODE 1
20 PRINT "EJEMPLO DEL NUEVO JUEGO DE CARACTERES"
30 PRINT "UTILIZANDO CALL &BA0A"
40 CALL &BA0A
```



Puede utilizarlo también en MODE 0 pero los caracteres son menos legibles.

23. CAT POR IMPRESORA CP/M!

664 y 6128

EFECTO

El efecto producido es idéntico al indicado en la ficha n.º 10. Enviando a la impresora el "catálogo" de todos los archivos contenidos en la cara del disco en curso.

CÓMO CONSEGUIRLO

Proceda de la siguiente forma:

- 1.º) Introduzca la cara 1 del disco CPM.
 - 2.º) Teclee CPM y ENTER.
 - 3.º) Cuando aparezca el "prompt" A>, teclee CONTROL + P. En la pantalla no escribirá nada y sonará un pitido de aviso.
 - 4.º) Saque el disco e introduzca el deseado.
 - 5.º) Teclee DIR (o dir) y ENTER.
- OJO no se olvide encender antes su impresora.

EJEMPLO

Siga los pasos indicados en el orden expuesto y ¡no olvide encender su impresora!

EXPLICACIÓN

Bajo el sistema operativo CP/M, CONTROL + P indica al ordenador que todas las salidas han de ser enviadas simultáneamente a la impresora y a la pantalla.

Si pulsamos nuevamente CONTROL + P, se cancela la salida a impresora, de modo que esta función es un conmutador de la impresora.